pragma solidity ^0.4.11;


/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {
  function mul(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a * b;
    assert(a == 0 || c / a == b);
    return c;
  }

  function div(uint256 a, uint256 b) internal constant returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  function sub(uint256 a, uint256 b) internal constant returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  function add(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}


/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() {
    owner = msg.sender;
  }


  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }


  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) onlyOwner {
    if (newOwner != address(0)) {
      owner = newOwner;
    }
  }

}

/**
 * @title Distributable
 * @dev The Distribution contract has multi dealer address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Distributable is Ownable {
  mapping(address => bool) public dealership;
  event Trust(address dealer);
  event Distrust(address dealer);

  modifier onlyDealers() {
    require(dealership[msg.sender]);
    _;
  }

  function trust(address newDealer) onlyOwner {
    require(newDealer != address(0));
    require(!dealership[newDealer]);
    dealership[newDealer] = true;
    Trust(newDealer);
  }

  function distrust(address dealer) onlyOwner {
    require(dealership[dealer]);
    dealership[dealer] = false;
    Distrust(dealer);
  }

}


contract Pausable is Ownable {
  event Pause();
  event Unpause();

  bool public paused = false;


  /**
   * @dev modifier to allow actions only when the contract IS paused
   */
  modifier whenNotPaused() {
    require(!paused);
    _;
  }

  /**
   * @dev modifier to allow actions only when the contract IS NOT paused
   */
  modifier whenPaused {
    require(paused);
    _;
  }

  /**
   * @dev called by the owner to pause, triggers stopped state
   */
  function pause() onlyOwner whenNotPaused returns (bool) {
    paused = true;
    Pause();
    return true;
  }

  /**
   * @dev called by the owner to unpause, returns to normal state
   */
  function unpause() onlyOwner whenPaused returns (bool) {
    paused = false;
    Unpause();
    return true;
  }
}




/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  uint256 public totalSupply;
  function balanceOf(address who) constant returns (uint256);
  function transfer(address to, uint256 value) returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}


/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) returns (bool) {
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) constant returns (uint256 balance) {
    return balances[_owner];
  }

}


/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) constant returns (uint256);
  function transferFrom(address from, address to, uint256 value) returns (bool);
  function approve(address spender, uint256 value) returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}


/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amout of tokens to be transfered
   */
  function transferFrom(address _from, address _to, uint256 _value) returns (bool) {
    var _allowance = allowed[_from][msg.sender];

    // Check is not needed because sub(_allowance, _value) will already throw if this condition is not met
    // require (_value <= _allowance);

    balances[_to] = balances[_to].add(_value);
    balances[_from] = balances[_from].sub(_value);
    allowed[_from][msg.sender] = _allowance.sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Aprove the passed address to spend the specified amount of tokens on behalf of msg.sender.
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) returns (bool) {

    // To change the approve amount you first have to reduce the addresses`
    //  allowance to zero by calling `approve(_spender, 0)` if it is not
    //  already 0 to mitigate the race condition described here:
    //  https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
    require((_value == 0) || (allowed[msg.sender][_spender] == 0));

    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifing the amount of tokens still avaible for the spender.
   */
  function allowance(address _owner, address _spender) constant returns (uint256 remaining) {
    return allowed[_owner][_spender];
  }

}


contract DistributionToken is StandardToken, Distributable {
  event Mint(address indexed dealer, address indexed to, uint256 value);
  event Burn(address indexed dealer, address indexed from, uint256 value);

   /**
   * @dev to mint tokens
   * @param _to The address that will recieve the minted tokens.
   * @param _value The amount of tokens to mint.
   * @return A boolean that indicates if the operation was successful.
   */
  function mint(address _to, uint256 _value) onlyDealers returns (bool) {
    totalSupply = totalSupply.add(_value);
    balances[_to] = balances[_to].add(_value);
    Mint(msg.sender, _to, _value);
    Transfer(address(0), _to, _value);
    return true;
  }


  function burn(address _from, uint256 _value) onlyDealers returns (bool) {
    totalSupply = totalSupply.sub(_value);
    balances[_from] = balances[_from].sub(_value);
    Burn(msg.sender, _from, _value);
    Transfer(_from, address(0), _value);
    return true;
  }
}


contract EverFountainBeanSale is Ownable, Pausable, Distributable {
  using SafeMath for uint256;
  event Sale(address indexed customer, uint256 value, uint256 amount, uint256 consume, string order, uint256 reward);
  struct FlexibleReward {
    uint256 percentage;
    uint256 limit;
  }

  uint256 public totalSales;
  uint256 public totalReward;
  uint256 public totalConsume;
  FlexibleReward[] public flexibleRewardLevel;
  uint256 flexibleRewardIndex = 0;

  // address where funds are collected
  address public wallet;

  // how many token units a buyer gets per wei
  uint256 public rate;

  uint256 public weiRaised;

  DistributionToken public token;

  function EverFountainBeanSale(DistributionToken _token, uint256 _rate, address _wallet){
    require(_token != address(0));
    require(_rate > 0);
    require(_wallet != address(0));
    token = _token;
    wallet = _wallet;
    rate = _rate;

    flexibleRewardLevel.push(FlexibleReward({ limit:1000000, percentage:15}));
    flexibleRewardLevel.push(FlexibleReward({ limit:3000000, percentage:13}));
    flexibleRewardLevel.push(FlexibleReward({ limit:6000000, percentage:11}));
    flexibleRewardLevel.push(FlexibleReward({ limit:10000000, percentage:9}));
    flexibleRewardLevel.push(FlexibleReward({ limit:15000000, percentage:7}));
    flexibleRewardLevel.push(FlexibleReward({ limit:21000000, percentage:5}));
    flexibleRewardLevel.push(FlexibleReward({ limit:0, percentage:0}));
    trust(msg.sender);
  }

  function balanceOf(address _owner) constant returns (uint256 balance) {
    return token.balanceOf(_owner);
  }

  function calcFlexibleReward(uint256 amount) constant returns (uint256 reward){
    FlexibleReward memory level = flexibleRewardLevel[flexibleRewardIndex];
    if (level.limit == 0) {
      return 0;
    }
    FlexibleReward memory nextLevel = flexibleRewardLevel[flexibleRewardIndex + 1];
    uint256 futureTotalSales = totalSales.add(amount);
    uint256 benefit;
    if (nextLevel.limit == 0) {
      if (level.limit >= futureTotalSales) {
        return amount.mul(level.percentage).div(100);
      }
      benefit = level.limit.sub(totalSales);
      return benefit.mul(level.percentage).div(100);
    }

    require(nextLevel.limit > futureTotalSales);

    if (level.limit >= futureTotalSales) {
      return amount.mul(level.percentage).div(100);
    }

    benefit = level.limit.sub(totalSales);
    uint256 nextBenefit = amount.sub(benefit);
    return benefit.mul(level.percentage).div(100).add(nextBenefit.mul(nextLevel.percentage).div(100));

  }

  function calcFixedReward(uint256 amount) constant returns (uint256 reward){
    uint256 less6000Reward = 0;
    uint256 less24000Percentage = 5;
    uint256 mostPercentage = 15;

    if (amount < 6000) {
      return less6000Reward;
    }

    if (amount < 24000) {
      return amount.mul(less24000Percentage).div(100);
    }

    return amount.mul(mostPercentage).div(100);
  }

  function calcReward(uint256 amount) constant returns (uint256 reward){
    return calcFixedReward(amount).add(calcFlexibleReward(amount));
  }

  function flexibleReward() constant returns (uint256 percentage, uint limit){
    FlexibleReward memory level = flexibleRewardLevel[flexibleRewardIndex];
    return (level.percentage, level.limit);
  }

  function nextFlexibleReward() constant returns (uint256 percentage, uint limit){
    FlexibleReward memory nextLevel = flexibleRewardLevel[flexibleRewardIndex+1];
    return (nextLevel.percentage, nextLevel.limit);
  }

  function setRate(uint256 _rate) onlyDealers returns(bool) {
    require(_rate > 0);
    rate = _rate;
    return true;
  }

  function destroy() onlyOwner {
    selfdestruct(owner);
  }

  function changeWallet(address _wallet) onlyOwner returns(bool) {
    require(_wallet != address(0));
    wallet = _wallet;
    return true;
  }

  function trade(uint256 amount, uint256 consume, string order) payable whenNotPaused returns(bool){
    require(bytes(order).length > 0);
    uint256 balance;
    if (msg.value == 0) {
      //only consume
      require(consume > 0);
      require(amount == 0);
      balance = token.balanceOf(msg.sender);
      require(balance >= consume);
      totalConsume = totalConsume.add(consume);
      token.burn(msg.sender, consume);
      Sale(msg.sender, msg.value, amount, consume, order, 0);
      return true;
    }

    require(amount > 0);
    uint256 sales = msg.value.div(rate);
    require(sales == amount);
    totalSales = totalSales.add(sales);
    uint256 reward = calcReward(sales);
    totalReward = totalReward.add(reward);
    FlexibleReward memory level = flexibleRewardLevel[flexibleRewardIndex];
    if (level.limit>0 && totalSales >= level.limit) {
      flexibleRewardIndex = flexibleRewardIndex + 1;
    }
    uint256 gain = sales.add(reward);

    if (consume == 0) {
      //only sale token
      token.mint(msg.sender, gain);

      weiRaised = weiRaised.add(msg.value);
      wallet.transfer(msg.value);

      Sale(msg.sender, msg.value, amount, consume, order, reward);
      return true;
    }

    balance = token.balanceOf(msg.sender);
    uint256 futureBalance = balance.add(gain);
    require(futureBalance >= consume);

    totalConsume = totalConsume.add(consume);
    token.mint(msg.sender, gain);
    token.burn(msg.sender, consume);

    weiRaised = weiRaised.add(msg.value);
    wallet.transfer(msg.value);

    Sale(msg.sender, msg.value, amount, consume, order, reward);
    return true;
  }

}